



NRL/MR/6390--10-9260

Initial Construction of a General Framework for Numerical Simulation of IED Detection and Remote Activation Scenarios

A. SHABAEV

*George Mason University
Fairfax, Virginia*

S.G. LAMBRAKOS

N. BERNSTEIN

V. JACOBS

*Center for Computational Materials Science
Materials Science and Technology Division*

D. FINKENSTADT

*U.S. Naval Academy
Annapolis, Maryland*

June 18, 2010

Approved for public release; distribution is unlimited.

Contents

Introduction.....	1
General Simulation Framework.....	2
Description Of Component Models.....	2
Prototype Analysis (THz Excitation).....	6
Analysis Of Permittivity Functions Using DFT.....	8
Conclusion.....	12
References.....	12
Disclaimer.....	13
Appendix 1: Computer Program For S-Matrix.....	14
Appendix 2: EM Response Calculations using DFT.....	26

Introduction

This report describes an initial construction of a general framework for numerical simulation of the various possible types of scenarios that could possibly occur for the detection and remote activation of improvised explosive devices (IEDs) by excitation of incident electromagnetic waves. This general framework consists of a set of component models, each of whose structure permits the output of given types of information. The model representing the central component of this framework, to which the outputs of all the other component models are inputs, is that of an S-matrix representation of a multilayered composite material system, where each layer of the system is characterized by an average thickness and effective electric permittivity function [1]. The outputs of this primary component are the reflectivity and transmissivity as a function of frequency and incident angle of the incident electromagnetic wave. The other component models, whose outputs are input to the S-matrix model, are response spectra calculated using density functional theory (DFT) [2-4] and related methodologies, parameterized analytic function representations of the electric permittivity as a function of frequency obtained by fitting experimentally measured spectra, and effective permittivity functions whose construction is based on effective medium theory (EMT) and roughness models. We review those physical theories establishing the foundation of the component models and a prototype simulation that considers response characteristics for THz excitation. We include an initial version of a computer program for calculation of reflectivity and transmissivity functions using the S-matrix formulation. Aspects of this specific software implementation are discussed. In addition, we describe a procedure for calculating response spectra using DFT for use as input to the S-matrix model. For this purpose we have adopted the DFT software NRLMOL.

It is significant to note that the numerical-simulation framework to be presented is structured for two major purposes, which are complimentary. One purpose, which relates directly to practical application, is simulation of various possible scenarios for detection of IEDs corresponding to the presence of various types of intermediate material layers between explosive and detector. The other purpose, which relates indirectly to practical application, but is yet extremely important for the interpretation and design of detection strategies, is the quantitative analysis of absolute bounds, or rather, the inherent limitation on levels of detection associated with various types of detection strategies. With respect to the purpose of examining inherent limitations on IED detection, the dominant features of response spectra that are calculated using DFT provide a foundation for establishing what level of detection is achievable in the absence of instrumental and environmental factors associated with detection. Accordingly, the simulation framework presented here considers a specific application of DFT. For any given energetic material and frequency range of the incident electromagnetic wave, the output of the DFT model component is a set of response signatures that are each characterized by an excitation frequency, magnitude and width. These response signatures must then be used to construct permittivity functions, which represent the form of input to the S-matrix component of the simulation framework.

A significant aspect of the numerical-simulation framework presented is that it adopts the perspective of computational physics, according to which a numerical




Figure 1. General framework for numerical simulation of IED detection and remote activation scenarios.

simulation represents another source of “experimental” data. This perspective is significant in that a general procedure may be developed for construction of permittivity functions that uses both DFT calculations as well as experimental measurements. That is to say, for the purpose of simulating many electromagnetic response characteristics of energetic materials, DFT and associated methodologies such as tight binding (TB) methods, are sufficiently mature for the purpose of generating data that complements experimental measurements rather than simply providing verification.

General Simulation Framework

A schematic representation of the general framework for numerical simulation of IED response is shown in Fig. 1. It should be emphasized that this represents an initial construction and that the general framework shown in Fig. 1 is subject to subsequent refinement and modifications with respect to the paths of input and output from the different model components comprising the framework. Referring to Fig. 1, it should be noted that the primary input to the model system is the set of permittivity functions that are associated with the different layers of material making the system.

Description Of Component Models

S-Matrix Representation of Layered Composite System

The central component of the general simulation framework, to which the outputs of all the other component models are inputs, is that of an S-matrix representation of a multilayered composite material system, where each layer of the system is characterized by an average thickness and effective electric permittivity function (see [5]). The outputs of this central component are the reflectivity and transmissivity as a function of frequency,

incident angle and polarization of the incident electromagnetic wave. The formulation of the S-matrix representation is defined by the following system of equations.

The reflectivity R and transmissivity T functions are given by

$$R = -\frac{S_{12}}{S_{11}} \quad \text{and} \quad T = \frac{S_{11}S_{22} - S_{12}S_{21}}{S_{11}}, \quad (1)$$

respectively, where the S-matrix elements S_{ii} are define by the matrix relation

$$[S] = \begin{pmatrix} S_{11} & S_{12} \\ S_{21} & S_{22} \end{pmatrix} = \prod_{j=1}^m [M_j] \quad (2)$$

where $[M_j] = [I_{(j-1)j}][L_j]$. The matrix $[I_{ab}]$ is defined by the matrix relation

$$[I_{ab}] = \frac{1}{t_{ab}} \begin{pmatrix} 1 & r_{ab} \\ r_{ab} & 1 \end{pmatrix} \quad (3)$$

where

$$r_{ab} = \frac{\varepsilon_b S_a - \varepsilon_a S_b}{\varepsilon_b S_a + \varepsilon_a S_b} \quad \text{and} \quad t_{ab} = \frac{2S_a}{S_a + S_b} \quad (4)$$

for a p-polarized incident wave, and

$$r_{ab} = \frac{S_a - S_b}{S_a + S_b} \quad \text{and} \quad t_{ab} = \frac{2\varepsilon_b S_a}{\varepsilon_b S_a + \varepsilon_a S_b} \quad (5)$$

for an s-polarized incident wave, where

$$S_a = (\varepsilon_a - \varepsilon_0 \sin^2 \phi)^{1/2} \quad \text{and} \quad S_b = (\varepsilon_b - \varepsilon_0 \sin^2 \phi)^{1/2} \quad (6)$$

Here, ε_0 is the permittivity function of the transparent ambient layer, ε_a and ε_b are the permittivity functions for layers “a” and “b,” respectively. The matrix $[L_j]$ is defined by the matrix relation

$$[L_j] = \begin{pmatrix} X_j^{1/2} & 0 \\ 0 & 1/X_j^{1/2} \end{pmatrix} \quad (7)$$

where

$$X_j = \exp\left[-2\pi i \left(\frac{d_j}{D_j(\phi)}\right)\right], \quad D_j(\phi) = \frac{1}{2} \left(\frac{\lambda}{S_j}\right), \quad S_j = (\varepsilon_j - \varepsilon_0 \sin^2 \phi)^{1/2} \quad (8)$$

and ϵ_j is the permittivity functions for layer “ j ”. The layer indexing used in Eqs.(1) through (8) is defined with reference to Fig. 2. A computer program for numerical implementation of Eqs.(1) through (8) is given in Appendix 1.




Figure 2. Schematic representation of layer indexing used in Eqs.(1) through (8).

Dielectric Permittivity Functions

The set of permittivity functions that are associated with the different layers comprising the layered composite system represent the primary input to the S-matrix model component. These functions are to be constructed in principle for a given material according to a “best fit” of available information associated with the electromagnetic response of that material. For a given material this information consists of data obtained from both experimental measurements, e.g., reflectivity and absorption measurements, and numerical simulations based on basic principles, e.g., DFT and TB calculations. It is significant to note that the best fit to the electromagnetic response of a given material will depend on the specific response signature characteristics of that material. Accordingly, from the perspective of numerical simulation, a best fit can be in the form of a tabulated functional dependence, as well as representations using analytical functions.

There are specific materials that are typically present in the ambient environment associated with IED detection as well as the detection of other types of materials, e.g., water and water vapor. Accordingly, the electromagnetic response characteristics of these materials have received a considerable analysis by many groups and are available. It follows that the permittivity functions of these materials should represent a permanent “data base” component of the general simulation framework. An example of the measurement of absorption coefficients of selected explosives that are covered by different types of materials (plastic, cotton and leather) are given in Ref.(6).

Density Functional Theory and Related Methodologies

The application of density functional theory (DFT) and related methodologies for the determination of electromagnetic response characteristics is important for the analysis of parameter sensitivity. That is to say, many characteristics of the electromagnetic response of a given material may not be detectable, or in general, not relevant for detection. Accordingly, sensitivity analyses concerning the electromagnetic response of layered composite systems can adopt the results of simulations using DFT, and related methodologies, to provide realistic limits on detectability that are independent of a specific system design for IED detection. In addition, analysis of parameter sensitivity based on atomistic response characteristics of a given material, obtained by DFT, provide for an “optimal” best fit of experimental measurements for the construction of permittivity functions. It follows that within the context of parameter sensitivity analysis, data obtained by means of DFT represents a true complement to data that has been obtained by means of experimental measurements.

Experimental Measurements

The dominant amount of information that is adopted for the construction of permittivity functions is obtained from experimental measurements of electromagnetic response characteristics. Some major issues associated with these constructions are that such experimental measurements typically involve bulk material response characteristics as well as measurement errors due to sample surface preparations and artifacts due to ambient environmental influences. These issues are significant in that the permittivity functions adopted as input are typically assumed as being associated with “pure” materials as well as representative of response characteristics on a small scale that may be typical of thin film type layers. As in the case of response characteristics that are determined via atomistic calculations, certain response features associated with response characteristics determined by experimental measurement may not be significant for the simulation of IED detection. That is to say, certain features such as the locations and amplitudes of response spectra may be significant for inclusion into model representations, while only a reasonable estimate of the widths may be necessary. It follows that sensitivity analysis for parameterizations of experimental measurements is as relevant as those associated with theoretical predictions. Such analysis is another application goal of the simulation framework presented here.

Effective Medium Theory (EMT), Equivalent Layers and Roughness Models

Consistent with the goal of determining absolute limitations on detectability of IEDs by means of electromagnetic excitation is the construction of models of material response that are representative of a general class of materials and detection scenarios, in contrast to models that would tend to be associated with a specific experimental arrangement in the laboratory. Accordingly, the concepts of an “effective medium” and “equivalent layer” are significant in that their consideration for model construction can provide quantitative bounds on detectability for a wide range of detection scenarios. In particular, these concepts can provide a foundation for the parametric representation of surface roughness and inhomogeneities on various spatial scales in the ambient environment.

The formal structure of all continuum effective medium theories (EMTs) are based on mixing rules that are functions of the different permittivities making up the composite material (See [7] and references therein for further discussion of EMTs).

The concept of an equivalent layer is based on the fact that a given range of different types of layer structures can have the same average response characteristics. Accordingly, a layer structure that is within this range can be represented by means of an equivalent layer whose construction does not require consideration of many details associated with its composition. A particular case of the application of the concept of an equivalent layer in combination with EMT is the construction of roughness models for the representation of rough surface structure.

Prototype Analysis (THz Excitation)

Presented in this section is a prototype simulation for demonstrating some aspects of the relationship between the various model components that comprise the general simulation framework. For this simulation the response of a layer of β -HMX to THz excitation is considered [8-10].

Shown in Fig. 3 are the real and imaginary parts of a permittivity function corresponding to the electromagnetic response of β -HMX to excitation within the THz range of frequencies. This permittivity function is significant in that it has been constructed using DFT calculations that have been calibrated with reference experimental measurements. Accordingly, the approach followed for construction of this permittivity function is that which has been adopted for the construction of permittivity functions within the simulation framework presented, i.e., a best fit to the combination of both theoretical calculations and experimental data.

Shown in Fig. 4 are reflectivity functions corresponding to s and p-polarization of the incident wave. The layered system consists of a layer of β -HMX upon a gold substrate. The reflectivity functions shown in Fig. 4, in principle, would represent the starting point for any study concerning absolute bounds on the detectability of β -HMX under different environmental conditions (i.e., surface layers and ambient environment) and detection scenarios.




Figure 3. Permittivity function of β -HMX for frequencies within THz range.




Figure 4. Reflectivity functions for a layer of β -HMX on a gold substrate.

Analysis Of Permittivity Functions Using DFT

The general approach of constructing permittivity functions according to the best fit of available data for given material corresponding to many different types of experimental measurements is not unprecedented and has been typically the dominant approach, e.g., the permittivity function of water. The general simulation framework presented here considers an extension of this approach in that calculations of electromagnetic response based on DFT and associated methodologies are also adopted as data for construction of permittivity functions. The inclusion of this type of information is significant for accessing what spectral response features at the molecular level are actually detectable with respect to a given set of detection parameters. Accordingly, permittivity functions having been constructed using DFT calculations provide a quantitative correlation between macroscopic material response and molecular structure. Within this context it is not important that the permittivity function be quantitatively accurate for the purpose of being adopted as input for system simulation. Rather, it is important that permittivity function be qualitatively accurate in terms of its general features for the purpose of sensitivity analysis, which is relevant for the assessment of absolute detectability of different types of molecular structure with respect to a given set of detection parameters. That is to say, permittivity functions that have been determined using DFT can provide a mechanistic interpretation of material response to electromagnetic excitation that could establish the well posedness of a given detection methodology for detection of specific molecular characteristics. Within the context of practical application, permittivity functions having been constructed according to the best fit of available data would be “correlated” with those obtained using DFT for proper interpretation of permittivity-function features. Subsequent to establishment of good correlation between DFT and experiment, DFT calculations can be adopted as constraints for the purpose of constructing permittivity functions, whose features are consistent with molecular level response, for adjustment relative to specific sets of either experimental data or additional molecular level information.




Figure 5. General procedure for construction of permittivity functions using DFT calculations.

The construction of permittivity functions using DFT calculations involves, however, an aspect that requires serious consideration. This aspect concerns the fact that a specific parametric function representation must be adopted. This significant aspect of constructing permittivity functions using DFT, and related methodologies, is shown

explicitly in Fig. 5. Accordingly, any parametric representation, i.e., parameterization, adopted for permittivity-function construction must be physically consistent with specific molecular response characteristics, while limiting the inclusion of feature characteristics that tend to mask response signatures that may be potentially detectable.

In principle, parameterizations are of two classes. One class consists of parameterizations that are directly related to molecular response characteristics. This class of parameterizations would include spectral scaling and width coefficients. The other class consists of parameterizations that are purely phenomenological and are structured for optimal and convenient best fits to experimental measurements.

At this stage it is instructive to present a prototype calculation demonstrating analysis, e.g., interpretation, of permittivity-function features using DFT calculations. Consistent with the prototype simulation presented above, a permittivity function is constructed using DFT calculations for β -HMX response to THz excitation. Shown in Fig. 6 is a general description of the geometry of the β -HMX molecule that was adopted for calculation of a permittivity function using DFT. That is to say, the molecular structure that was input to the DFT software NRLMOL. Shown in Figs. 7 and 8 are absorption coefficients corresponding to different adjustable spectral scaling and width parameters.




Figure 6. Molecular structure of β -HMX used for DFT calculations of spectral response.

Remark. It is significant to note that with respect to practical application the absorption coefficient α

$$\alpha = \frac{4\pi}{\sqrt{2}\lambda} \left[-\varepsilon_r + \sqrt{\varepsilon_r^2 + \varepsilon_i^2} \right]^{1/2} \quad (9)$$

where λ , ε_r and ε_i are the wavelength of excitation, and the real and imaginary parts of the permittivity function, respectively, provides a direct relationship between a calculated quantity using DFT and a “conveniently measurable” quantity α .

Next, we consider a qualitative example of examining the correlation between DFT calculated permittivity functions and experimental measurements. Referring to Fig. 9, which shows an experimentally determined absorption coefficient for β -HMX (see Ref.(7)), we note good correlation between the permittivity functions (in terms of their α representation) obtained by DFT (using NRLMOL) and experiment. Most importantly, the level of correlation is sufficient to establish a “proof of concept” that DFT calculations provide a quantitative initial estimate of molecular response to electromagnetic excitation for subsequent parameterization [11-13].




Figure 7. Absorption coefficient for β -HMX calculated by DFT for THz range of frequencies corresponding to adjustable parameters $\gamma = 3 \text{ cm}^{-1}$ and $N = 2 \text{ cm}^{-2}$.

As indicated previously, the parameterizations applied to DFT calculations will in general consist of two classes of parameterizations, i.e., one consistent with basic theory and the other consistent with optimal and convenient best fitting of experimental measurements. Accordingly, one class of parameterization defines a problem requiring further analysis in terms of basic theory [8,10,14,15], while the other class defines a problem requiring analysis in terms of inverse-problem and parameter-optimization methodologies [16].




Figure 8. Absorption coefficient for β -HMX calculated by DFT for THz range of frequencies corresponding to adjustable parameters $\gamma = 5 \text{ cm}^{-1}$ and $N = 3 \text{ cm}^{-2}$.




Figure 9. Experimentally determined absorption coefficient for β -HMX for THz range of frequencies (see [8]).

Conclusion

The development of IED detection methodologies requires the consideration of two major aspects of detection. These are the detectability of IEDs under different types of environmental conditions and detection scenarios; and the absolute detectability of the different types of response characteristics of energetic materials due to electromagnetic wave excitation. Accordingly, within the context of practical application of IED detection methodologies, it remains necessary to establish correlation with response properties on the molecular level. It is therefore necessary to construct two types of permittivity functions. One type, whose purpose is the simulation of detection scenarios, represents the best fit to available data, which could include both experimental measurements and calculations based on theory. The other type, obtained using DFT, is that of a reasonably optimal parametric representation of molecular level response characteristics, providing interpretation of permittivity-function features at the molecular level, whose purpose is that of initial constraints for subsequent adjustment relative to specific sets of either experimental data or additional molecular level information. It follows that the establishment of a general constrained parameterization of IED response based on both theory and experiment, combined with quantitative sensitivity analyses of this parameterization, will provide an assessment of the general detectability of IEDs independent of specific detection scenarios.

Acknowledgement

This work was supported by the Office of Naval Research. The authors would like to thank Dr. R. Guirguis, Dr. J. Hooper and Dr. J. Wilkinson of the Naval Surface Warfare Center, Indian Head Division, for their discussions during this effort. In addition, the authors would like to thank Dr. M. R. Pederson for his discussions concerning NRLMOL.

References

1. S. Jaruwatanadilok, Y. Kuga, A. Ishimaru, and E. Thorsos, "An electromagnetic model for detecting explosives under obscuring layers" Intern. Symp. On Technology and the Mine problem, Monterey, CA, May 2-4, 2006.
2. P. Hohenberg and W. Kohn, "Inhomogeneous Electron Gas" Phys. Rev. **136**, B864, (1964).
3. W. Kohn and L. J. Sham, "Self-Consistent Equations Including Exchange and Correlation Effects" Phys. Rev. **140**, A1133 (1965).
4. R.O. Jones and O. Gunnarson, "The density functional formalism, its applications and prospects" Rev. Mod. Phys. **61**, 689 (1989).
5. L. D. Landau, L. P. Pitaevskii, and E. M. Lifshitz, *Electrodynamics of Continuous Media* (Elsevier Butterworth-Heinemann, Oxford-Burlington, 2004).

6. J. Chen, Y. Chen, H. Zhao, G. J. Bastiaans, and X.-C. Zhang, "Absorption coefficients of selected explosives and related compounds in the range of 0.1-2.8 THz," *Optics Express* **15**, 11763 (2007).
7. C. F. Bohren and D. R. Huffman, *Absorption and Scattering of Light by Small Particles* (Wiley-VCH Verlag, Weinheim, 2004).
8. J. Hooper, E. Mitchell, C. Konek, and J. Wilkinson, "Terahertz optical properties of the high explosive β -HMX" *Chem. Phys. Lett.* **467**, 309 (2009).
9. A. Häußler, T. M. Klapötke, G. Holl, and M. Kaiser, "A Combined Experimental and Theoretical Study of HMX (Octogen, Octahydro-1,3,5,7-Tetranitro-1,3,5,7-Tetrazocine) in the Gas Phase" *Propellants, Explosives, Pyrotechnics* **27**, 12 (2002).
10. D. G. Allis, D. A. Prokhorova, and T. M. Korter, "Solid-State Modeling of the Terahertz Spectrum of the High Explosive HMX" *J. Phys. Chem. A* **110**, 1951 (2006).
11. D. V. Porezag and M. R. Pederson, "Infrared intensities and Raman-scattering activities within density-functional theory" *Phys. Rev. B* **54**, 7830 (1996).
12. K. Jackson, M. R. Pederson, D. Porezag, Z. Hajnal, and T. Frauenheim, "Density-functional-based predictions of Raman and IR spectra for small Si clusters" *Phys. Rev. B* **55**, 2549 (1997).
13. E. B. Wilson, J. C. Decius, and P. C. Cross, *Molecular Vibrations* (McGraw-Hill, New York, 1955).
14. B. M. Rice and C. F. Chabalowski, "Ab Initio and Nonlocal Density Functional Study of 1,3,5-Trinitro-s-triazine (RDX) Conformers," *J. Phys. Chem. A* **101**, 8720 (1997).
15. D. G. Allis and T. M. Korter, "Theoretical Analysis of the Terahertz Spectrum of the High Explosive PETN," *ChemPhysChem* **7**, 2398 (2006).
16. S. G. Lambrakos, P. G. Moore, P. Loschialpo, and D. Smith, "Estimates of Optical Parameters Based on Mixing Rules Extending over a Wide Range of Wavelengths," *Appl. Spectr.*, **55**, 584 (2001) .

Disclaimer

This report does not imply any form of warranty that the S-Marix code does not contain errors or that it is sufficient for any specific application. The S-Matrix code should not be relied on for solving problems whose incorrect solution could result in damages. Accordingly, the authors of this S-Matrix code and this report disclaim all liability for direct or consequential damages resulting from the use of the S-Matrix code.

Appendix 1

Computer Program For Calculation Of S-Matrix

Presented in this section is an initial version of a computer program, i.e., the “S-Matrix code” (in Fortran 77) for calculation of the reflectivity and transmissivity function using the S-Matrix representation of a multilayered composite material system.

```
PROGRAM MLAYERSP
C
C version 21
C last time modified: 03/23/2010
C

implicit none

C parameters and constants
integer ML,LNR,NMP ! max number of layers, resonances, mesh points
parameter (ML = 20, LNR = 100, NMP = 10000)

real*8 c,pi ! speed of light and pi
parameter (c = 2.99792458d8, pi=3.14159265358979323846d0)

complex*16 SP(2,2),SS(2,2) ! scattering matrix

real*8 theta(NMP),angle(NMP) ! incident angle
real*8 ang0,ang1 ! and its range

real*8 wnum0,wnum1 ! range of wave numbers
real*8 omega(NMP),lamda(NMP) ! angular frequency and wave length
real*8 wnum(NMP),awnum(NMP) ! wavenumber and angular wavenumber

complex*16 E0(NMP),E(ML,NMP),ERead(NMP)

real*8 D(ML) ! thickness of each layer

complex*16 REFP,REFS,TRANP,TRANS,DETSS,DETSP
real*8 ref1b,ref2b
real*8 ref1(NMP),ref2(NMP)

integer IUNITS ! input units

integer ISCAN ! type of scan

integer NLAYER,NFACE,NREGION ! number of layers, faces, regions

integer fol ! first opaque layer

integer NMESHA,NMESHF,NBINA,NBINF ! number of angles, freq, binsize
```

```

integer NBINA2,NBINF2 ! half of binsize

integer IL,I,J,IB,JB,ILS ! counters

character*79 layerfl ! layer data file
integer ecode ! errorcode

open(1,file='MLAYER.INP') ! open input and output files
open(2,file='MLAYER.OUT')
open(3,file='MLAYER.AVG')

read(1,*) IUNITS, ISCAN ! input units and type of scan
read(1,*) wnum0,wnum1,NMESHF,NBINF ! wave numbers
NBINF=2*INT((NBINF+1)/2)-1
NBINF2=INT(NBINF/2)
do J=1,NMESHF+NBINF-1
  wnum(J) = wnum0 + (float(J-1)/float(NMESHF))*(wnum1-wnum0)
  lamda(J)=1.0d0/wnum(J) ! wave number
  awnum(J)=2.0d0*pi*wnum(J) ! angular wave number
  if (IUNITS .eq. 0) then
    omega(J) = 1.0d2*awnum(J)*c
  else
    omega(J) = awnum(J)*c
  end if
end do
read(1,*) ang0,ang1,NMESHA,NBINA ! angles
NBINA=2*INT((NBINA+1)/2)-1
NBINA2=INT(NBINA/2)
do I=1,NMESHA+NBINA-1
  theta(I) = ang0 + (float(I-1)/float(NMESHA))*(ang1-ang0)
  theta(I) = (theta(I)/1.8d2)*pi
  angle(I) = (theta(I)/pi)*1.8d2
end do

read(1,*) NLAYER ! number of layers
NFACE = NLAYER + 1 ! number of interfaces
NREGION = NLAYER + 2 ! number of regions

read(1,*) ! blank line
read(1,'(a79)') layerfl

call READLAYER(layerfl,NMESHF,wnum,EREAD,ecode) ! ambient layer
if (ecode .eq. 0) then
  do J=1,NMESHF
    E0(J) = EREAD(J)
  end do
else
  print *, 'INPUT ERROR = ',ecode
  print *, 'CANNOT READ AMBIENT LAYER'
  stop
end if

do IL=1,NFACE ! read layers
  print *, 'LAYER=',IL
  read(1,*) ! blank line
  read(1,'(a79)') layerfl ! layer data file name

```

```

read(1,*) D(IL) ! thickness
call READLAYER(layerfl,NMESHF,wnum,EREAD,ecode)
if (ecode .eq. 0) then
  do J=1,NMESHF
    E(IL,J) = EREAD(J)
  end do
else
  print *, 'INPUT ERROR = ',ecode
  print *, 'CANNOT READ LAYER = ',IL
  stop
end if
if (D(IL) .lt. 4.0d0/wnum1) then
  print *, 'WARNING! thickness is too small'
end if
end do ! end of layers

if (ISCAN .eq. 0) then ! angle scan
write(2,*) '      ANGLE          Rp          Rs '
do I = 1,NMESHA+NBINA-1
  ref1b=0.0d0
  ref2b=0.0d0
  do J = 1,NBINF
    ILS=0 ! set control layer number ILS and fol to 0
    fol=0
    do IL = 1,NFACE
      if (ILS .lt. NFACE) then ! call layers if JS < NFACE
        fol=IL
        if (IL .eq. 1) then ! initial call for ambient layer
          call LAYER(IL,NLAYER,E0,E0,E(IL,J),theta(I),
&             lamda(J),0.0d0,SP,SS,ILS)
        else
          call LAYER(IL,NLAYER,E0,E(IL-1,J),E(IL,J),theta(I),
&             lamda(J),D(IL-1),SP,SS,ILS)
        end if ! IL=1
      end if ! ILS<NFACE
    end do ! end of layers
    REFP = -SP(1,2)/SP(1,1)
    DETSP=SP(1,1)*SP(2,2)-SP(1,2)*SP(2,1)
    TRANP = DETSP/SP(1,1)
    REFS = -SS(2,1)/SS(1,1)
    DETSS=SS(1,1)*SS(2,2)-SS(1,2)*SS(2,1)
    TRANS = DETSS/SS(1,1)
    ref1b=ref1b+REFP*conjg(REFP)
    ref2b=ref1b+REFS*conjg(REFS)
  end do ! end of frequency bin
  ref1(I)=ref1b/NBINF
  ref2(I)=ref2b/NBINF
  write(2,*) angle(I), ref1(I), ref2(I)
end do ! end of angle scan
write(3,*) '      ANGLE          Rp          Rs '
do I = 1,NMESHA
  ref1b=0.0d0
  ref2b=0.0d0
  do IB = 1,NBINA
    ref1b=ref1b+ref1(I-NBINA2+IB-1)
    ref2b=ref2b+ref2(I-NBINA2+IB-1)
  end do

```

```

    reflb=reflb/NBINA
    ref2b=ref2b/NBINA
    write(3,*) angle(J), reflb, ref2b
end do

else ! frequency scan
write(2,*) ' wnum      frequency      Rp      Rs '
do J = 1,NMESHF+NBINF-1
    reflb=0.0d0
    ref2b=0.0d0
    do I = 1,NBINA
        ILS=0 ! set control layer number JS and fol to 0
        fol=0
        do IL = 1,NFACE
            if (ILS .lt. NFACE) then ! call layers if ILS < NFACE
                fol=IL
                if (IL .eq. 1) then
                    call LAYER(IL,NLAYER,E0,E0,E(IL,J),theta(I),
&                        lamda(J),0.0d0,SP,SS,ILS)
                else
                    call LAYER(IL,NLAYER,E0,E(IL-1,J),E(IL,J),theta(I),
&                        lamda(J),D(IL-1),SP,SS,ILS)
                end if ! IL=1
            end if ! JS<NFACE
        end do ! end of layers
        REFP = SP(1,2)/SP(1,1)
        DETSP=SP(1,1)*SP(2,2)-SP(1,2)*SP(2,1)
        TRANP = DETSP/SP(1,1)
        REFS = SS(1,2)/SS(1,1)
        DETSS=SS(1,1)*SS(2,2)-SS(1,2)*SS(2,1)
        TRANS = DETSS/SS(1,1)
        reflb=reflb+REFP*conjg(REFP)
        ref2b=ref2b+REFS*conjg(REFS)
    end do ! end of angle bin
    ref1(J)=reflb/NBINA
    ref2(J)=ref2b/NBINA
    write(2,*) wnum(J), omega(J), ref1(J), ref2(J), fol
end do ! end of frequency scan
write(3,*) ' wnum      frequency      Rp      Rs '
do J = 1,NMESHF
    reflb=0.0d0
    ref2b=0.0d0
    do JB = 1,NBINF
        reflb=reflb+ref1(J-NBINF2+JB-1)
        ref2b=ref2b+ref2(J-NBINF2+JB-1)
    end do
    reflb=reflb/NBINF
    ref2b=ref2b/NBINF
    write(3,*) wnum(J), omega(J), reflb, ref2b
end do
end if ! angle or frequency scan

close(1) ! close input/output
close(2)
close(3)

stop

```

end ! end of program

C-----SUBROUTINES-----

```
subroutine NIM2ALPHA(nim,k,alpha)
C converts absorption coefficient to imaginary part of refraction index
implicit none
real*8 nim,alpha,k ! im part of ref index, abs coef, ang wavenumber
alpha = (2.0d0*k)*nim
return
end

subroutine ALPHA2NIM(alpha,k,nim)
C converts absorption coefficient to imaginary part of refraction index
implicit none
real*8 nim,alpha,k ! im part of of ref index, abs coef, ang wavenumber
nim = alpha/(2.0d0*k)
return
end

subroutine N2E(nre,nim,ere,eim)
C converts index of refraction to dielectric constant
implicit none
real*8 nre,nim,ere,eim
ere = nre**2 - nim**2
eim = 2.0d0*nre*nim
return
end

subroutine E2N(nre,nim,ere,eim)
C converts index of refraction to dielectric constant
implicit none
real*8 sqrt
intrinsic sqrt
real*8 nre,nim,ere,eim
nre = sqrt(ere + sqrt(ere**2 + eim**2))/sqrt(2.0d0)
nim = sqrt(-ere + sqrt(ere**2 + eim**2))/sqrt(2.0d0)
return
end

subroutine SUMOFRES(ere,eim,einf,vol,wnum,
& wnumN,sxN,syN,szN,gN,NRES)
C calculates dielectric constants
implicit none
real*8 sqrt
intrinsic sqrt
real*8 pi
parameter (pi=3.14159265358979323846d0)
real*8 ere,eim,einf,vol,wnum,wnumCM
real*8 wnumN(*),sxN(*),syN(*),szN(*),gN(*)
real*8 stot,pref
integer NRES, I
```

```

    pref=(4*pi/vol)
    ere=einf
    eim=0.0d0
    do I=1,NRES
        stot=(sxN(I)+syN(I)+szN(I))/3.0d0
        ere = ere + pref*stot*(wnumN(I)**2-wnum**2)/
&          ((wnum**2-wnumN(I)**2)**2+gN(I)**2*wnum**2)
        eim = eim + pref*stot*(wnum*gN(I))/
&          ((wnum**2-wnumN(I)**2)**2+gN(I)**2*wnum**2)
    end do
    return
end

subroutine LAYER(JL,JN,E0,eIN,eOUT,theta,lamda,D,SP,SS,JS)
C----- calculates reflection and transmission
C      of em wave through a single layer interface
C      and accumulative reflection and transmission

C      layer k-1
C      ----- interface
C      layer k

C-----input/output parameters
C
C      JL - layer number (= k)
C      JN - number of layers (NLAYER)
C      E0 - dielectric constant of vacuum (abandoned in ver. 6)
C      eIN - dielectric constant of layer k-1
C      eOUT - dielectric constant of layer k
C      theta - incident angle
C      lamda - wavelength
C      D - thickness of layer k
C      SP,SS - scattering matrices
C      JS - flag (=JL for transparent layers) set to NFACE for opaque layer
C
C-----internal paramaters
C
C      RP,RS,TP,TS - complex reflection and transmission coefficients
C      MS,MP - layer k matrices
C      IP,IS - interface matrices
C      NFACE - NLAYER+1
C      OD - optical density of layer
C      JT - flag (=1 for transparent layer) set to 0 for opaque layer

    implicit none
    real*8 pi
    parameter (pi=3.14159265358979323846d0)

    COMPLEX*16 e0,eIN,eOUT
    real*8 theta,lamda,D

    complex*16 RP,RS,TP,TS ! complex ref and trans coefficients

    complex*16 MS(2,2), MP(2,2) ! layer matrices
    complex*16 L(2,2)

    complex*16 IP(2,2),IS(2,2)

```



```

complex*16 SP(2,2), SS(2,2) ! scattering matrices

C   complex*16 SA, SB, EA, EB, DB, DJ, XJ, YJ
    complex*16 SA, SB, EA, EB, DA, DJ, XJ, YJ
    complex*16 tSP(2,2),tSS(2,2)
    complex*16 iunit, A(2,2)

complex*16 OD ! optical density

integer JL,JN,JS,NFACE,JT

real*8 exp,sqrt,sin,dbble ! external functions
intrinsic exp,sqrt,sin,dbble
COMPLEX*16 cdsqrt,cdexp,dcmplx
intrinsic cdsqrt,cdexp,dcmplx

C   initialize tSP/tSS with input matrix
    tSP(1,1) = SP(1,1)
    tSP(2,2) = SP(2,2)
    tSP(1,2) = SP(1,2)
    tSP(2,1) = SP(2,1)
    tSS(1,1) = SS(1,1)
    tSS(2,2) = SS(2,2)
    tSS(1,2) = SS(1,2)
    tSS(2,1) = SS(2,1)

C   set JS and JT flags to 'transparent' values
    JS=JL
    JT=1

C   initilaize internal parameters
    NFACE= JN + 1
    iunit = dcplx(0.0d0,1.0d0)
    EA = eIN
    EB = eOUT
    SA = cdsqrt(EA - E0*(sin(theta))**2)
    SB = cdsqrt(EB - E0*(sin(theta))**2)

C   RP = (EB*SA - EA*SB)/(EB*SA + EA*SB)
    RS = (SA - SB)/(SA + SB)
    TP = 2.*EB*SA/(EB*SA + EA*SB)
    TS = 2.*SA/(SA + SB)

C   Calculate IP11(1), IP12(1), IP21(1), IP22(1)
    IP(1,1) = 1./TP
    IP(1,2) = RP/TP
    IP(2,1) = IP(1,2)
    IP(2,2) = IP(1,1)

C   Calculate IS11(1), IS12(1), IS21(1), IS22(1)
    IS(1,1) = 1./TS
    IS(1,2) = RS/TS
    IS(2,1) = IS(1,2)
    IS(2,2) = IS(1,1)
    DA = dcplx(D,0.0d0)
    DJ = (lamda/2.)*(1./SA)
    OD = -2.*iunit*pi*(DA/DJ)

C

```

```

C
if (DBLE(OD) .gt. 1.0d2) then ! TEST OPTICAL DENSITY
  JS=NFACE
  JT=0
end if
if (JL .eq. 1) then ! generate initial SP and SS
  tSP(1,1) = dcplx(1.0d0,0.0d0)
  tSP(2,2) = dcplx(1.0d0,0.0d0)*JT
  tSP(1,2) = dcplx(0.0d0,0.0d0)
  tSP(2,1) = dcplx(0.0d0,0.0d0)
  tSS(1,1) = dcplx(1.0d0,0.0d0)
  tSS(2,2) = dcplx(1.0d0,0.0d0)
  tSS(1,2) = dcplx(0.0d0,0.0d0)
  tSS(2,1) = dcplx(0.0d0,0.0d0)
end if

if (JT .eq. 1) then ! 'optically thin' layer
C Calculate L11(1), L12(1), L21(1), L22(1)
  YJ = cdexp(0.5d0*OD)
C Calculate propagation matrix
  L(1,1) = YJ
  L(2,2) = 1./YJ
  L(1,2) = dcplx(0.0d0,0.0d0)
  L(2,1) = dcplx(0.0d0,0.0d0)
C Calculate MP11(1), MP12(1), MP21(1), MP22(1)
  MP(1,1) = IP(1,1)*L(1,1) + IP(1,2)*L(2,1)
  MP(2,1) = IP(2,1)*L(1,1) + IP(2,2)*L(2,1)
  MP(1,2) = IP(1,1)*L(1,2) + IP(1,2)*L(2,2)
  MP(2,2) = IP(2,1)*L(1,2) + IP(2,2)*L(2,2)
C Calculate MS11(1), MS12(1), MS21(1), MS22(1)
  MS(1,1) = IS(1,1)*L(1,1) + IS(1,2)*L(2,1)
  MS(2,1) = IS(2,1)*L(1,1) + IS(2,2)*L(2,1)
  MS(1,2) = IS(1,1)*L(1,2) + IS(1,2)*L(2,2)
  MS(2,2) = IS(2,1)*L(1,2) + IS(2,2)*L(2,2)
C assign MP matrix to em matrix
  A(1,1) = MP(1,1)*tSP(1,1)+MP(1,2)*tSP(2,1)
  A(2,1) = MP(2,1)*tSP(1,1)+MP(2,2)*tSP(2,1)
  A(1,2) = MP(1,1)*tSP(1,2)+MP(1,2)*tSP(2,2)
  A(2,2) = MP(2,1)*tSP(1,2)+MP(2,2)*tSP(2,2)
  tSP(1,1) = A(1,1)
  tSP(1,2) = A(1,2)
  tSP(2,1) = A(2,1)
  tSP(2,2) = A(2,2)
C assign MS matrix to em matrix
  A(1,1) = MS(1,1)*tSS(1,1)+MS(1,2)*tSS(2,1)
  A(1,2) = MS(1,1)*tSS(1,2)+MS(1,2)*tSS(2,2)
  A(2,1) = MS(2,1)*tSS(1,1)+MS(2,2)*tSS(2,1)
  A(2,2) = MS(2,1)*tSS(1,2)+MS(2,2)*tSS(2,2)
  tSS(1,1) = A(1,1)
  tSS(1,2) = A(1,2)
  tSS(2,1) = A(2,1)
  tSS(2,2) = A(2,2)
end if
C Assign SP scattering matrix
  SP(1,1) = tSP(1,1)
  SP(2,1) = tSP(2,1)

```

```

SP(1,2) = tSP(1,2)
SP(2,2) = tSP(2,2)
C   Assign SS scattering matrix
SS(1,1) = tSS(1,1)
SS(2,1) = tSS(2,1)
SS(1,2) = tSS(1,2)
SS(2,2) = tSS(2,2)
C
return
end      ! end of subroutine

subroutine READLAYER(layerfl,NMESH,wnum,eOUT,ecode)
C
C   reads layer parameters from data file
C
implicit none
real*8 pi
parameter (pi=3.14159265358979323846d0)
integer LNR ! max number of resonances
parameter (LNR = 100)
integer INTYPE ! layer input type wnum(I)
integer NMESH ! number of frequencies
integer NRES ! number of resonances
real*8 wnumN(LNR),sxN(LNR),syN(LNR),szN(LNR),gN(LNR) ! resonances
real*8 EINF,CVOL ! dielectric constant at infinity and cell volume
real*8 NR,NI ! index of refraction (NR,NI)
real*8 ER,EI ! permittivity E = (ER,EI)
real*8 ERT1,ERT2,EIT1,EIT2 ! permittivity from input table wnum(I)
real*8 ERTD,EITD ! permittivity difference in input table
real*8 wnumT1,wnumT2 ! wave numbers from input table
real*8 wnumTD ! wave numbers difference in input table
integer NTAB ! number of entries in data table
integer NTABERR ! error flag for mesh/data table mismatch
real*8 wnum(*) ! wavenumber
integer IL,I,J,JS ! counters
COMPLEX*16 eOUT(*)
character*79 layerfl ! layer data file
integer ecode ! errorcode

COMPLEX*16 dcmplx
intrinsic dcmplx

ecode = 0
open(unit=10,file=layerfl,status='old',iostat=ecode) ! layer data
if (ecode .eq. 0) then
  read(10,*) INTYPE ! ambient region input type
  if (INTYPE .eq. 0) then
    read(10,*) NR, NI ! index of refraction
    call N2E(NR,NI,ER,EI)
    do I=1,NMESH
      eOUT(I) = dcmplx(ER,EI)
    end do
  elseif (INTYPE .eq. 1) then ! dielectric constant
    read(10,*) ER, EI
    do I=1,NMESH
      eOUT(I) = dcmplx(ER,EI)
    end do
  end if
end if

```

```

    end do
elseif (INTYPE .eq. 2) then ! resonances
    read(10,*) EINF, CVOL
    read(10,*) NRES
    print *, 'NRES=',NRES
    do J=1,NRES
        read(10,*) wnumN(J),sxN(J),syN(J),szN(J),gN(J)
    end do
    call SUMOFRES(ER,EI,EINF,CVOL,wnum,
&                wnumN,sxN,syN,szN,gN,NRES)
    do I=1,NMESH
        eOUT(I) = dcplx(ER,EI)
    end do
elseif (INTYPE .eq. 3) then ! table
    do I=1,NMESH
        if (ecode .ge. 0) then
            NTABERR = 1
            close(10)
            open(unit=10,file=layerfl,status='old',iostat=ecode)
            if (ecode .eq. 0) then
                read(10,*) INTYPE ! input type
                read(10,*) NTAB ! number of entry lines
                JS = 2 ! read first two lines
                read(10,*) wnumT1, ERT1, EIT1
                read(10,*) wnumT2, ERT2, EIT2
                do J=3,NTAB
                    if ((wnum(I) .lt. wnumT1) .or.
&                    (wnum(I) .gt. wnumT2)) then
                        wnumT1 = wnumT2
                        ERT1 = ERT2
                        EIT1 = EIT2
                        JS= JS + 1 ! increase line number by 1
                        read(10,*) wnumT2, ERT2, EIT2
                    end if
                end do
                if (.not. (((wnum(I) .lt. wnumT1) .or.
&                    (wnum(I) .gt. wnumT2)))) then
                    wnumTD = wnumT2 - wnumT1
                    ERTD = ERT2 - ERT1
                    EITD = EIT2 - EIT1
                    if (wnumTD .gt. 1d-300) then
                        ER = ERT1 + ERTD*(wnum(I) - wnumT1)/wnumTD
                        EI = EIT1 + EITD*(wnum(I) - wnumT1)/wnumTD
                        NTABERR = 0
                    else
                        ER = ERT1
                        EI = EIT1
                        NTABERR = 2
                    end if
                end if
                eOUT(I) = dcplx(ER,EI)
                if (NTABERR .eq. 1) then
                    ecode=2000
                    print *, 'INPUT ERROR: mesh/table mismatch'
                end if
            end if ! ecode = 0
        end if ! ecode >= 0
    end if ! ecode >= 0

```

```

        end do
    end if ! INTYPE
end if ! ecode = 0 on input
close(10) ! close layer data
return
end ! end of subroutine

```

Brief Tutorial

1. Setup the geometry and composition of layers in input file (MLAYER.INP).

Example:

```

0          1
50.0      150.0d0    1001    1
40.0d0    90.0d0    101     1
2
-----ambient layer
data/AIR
-----layer 1
data/HMX.table
1.0d0
-----layer 2
data/GOLD
54.5d-1
-----layer 3
data/AIR
1.00d0

```

The first line is for units (0 – CGS; 1 - SI) and scan type (0 – angle scan; 1 – wavenumber scan).

The second and the third lines contain the range of scan, the number of mesh points and the bin size: the second line is for the wavenumber scan and the third line is for the angle scan.

The fourth line contains the number of layers (excluding ambient layer and substrate).

The fifth line is a divider.

The sixth line is for the path and the file name of the data file of the ambient layer.

The sixth line is a divider.

The seventh line is for the path and the file name of the data file of the first layer.

The eighth line contains the thickness of the first layer.

The last sequence of lines (from the sixth to the eighth is repeated as many time as many layers are declared on the fourth line plus one extra set for the substrate (layer 3 in our example).

2. Place data files for each layer at the location specified in MLAYER.INP file.

Examples of data files:

AIR file

0
1.0d0 0.0d0

The first line is for input type:

- 0 – real and imaginary parts of the refraction index
- 2 – table of resonances (frequency, oscillator strength, broadening)
- 3 – real and imaginary parts of the permittivity in the form of data table

HMX.table file

3
201
50.0000000000000 4.25905358483179 0.124136870938063
50.4975124262273 4.28289690235066 0.131953983279890
50.9950248524547 4.30819579494899 0.140717478521852
51.4925372786820 4.33508465558413 0.150560832568167
51.9900497049093 4.36371943716915 0.161650246864707
52.4875622242689 4.39427844001136 0.174192141418067
52.9850745573640 4.42696385313709 0.188442912146539
53.4825868904591 4.46200369947143 0.204721994922112
53.9800994098186 4.49965349284311 0.223429373616938
54.4776119291782 4.54019690996128 0.245069178213272
.
.
.
.
.
.
146.019899845123 2.33534257788569 3.796482250904258E-002
146.517413854599 2.34274505049977 3.704172732575566E-002
147.014927864075 2.34997051451085 3.615343225924952E-002
147.512435913086 2.35702540917843 3.529812379956010E-002
148.009949922562 2.36391609417684 3.447408114083763E-002
148.507463932037 2.37064836425402 3.367972744632545E-002
149.004977941513 2.37722780923246 3.291357929336934E-002
149.502485990525 2.38365967068105 3.217425750164293E-002

The first line is for input type (3 - data table)
The second line contains the number of data points in the table.
The table contains
wavenumber (first column),
permittivity: real (second column) and imaginary (second column) parts.

HMX file

```
2
2.9d0      1.0d0
8
46.5d0     0.00d0    0.08d0    0.00d0    5.00d0
62.9d0     0.00d0    61.54d0   0.00d0    5.00d0
76.3d0     10.36d0    0.00d0    5.22d0    5.00d0
84.2d0     30.71d0    0.00d0    29.97d0   5.00d0
87.9d0     0.00d0    79.15d0   0.00d0    5.00d0
97.8d0     0.00d0    29.38d0   0.00d0    5.00d0
100.3d0    4.75d0    0.00d0    204.88d0  5.00d0
116.0d0    57.01d0   0.00d0    68.39d0   5.00d0
```

The first line is for input type (2 - table of resonances).

The second column contains high-frequency permittivity and the volume of the unit cell.

The third line contains the number of data points in the table.

The table contains

wavenumber (first column),

oscillator strength: x (second column), y (third column), and z (fourth column) components,

broadening of each resonance

Appendix 2

EM Response Calculations using DFT

Presented in this section is a brief tutorial that describes the sequences of procedures for calculation of response spectra using density functional theory (DFT). The simulation framework presented here adopts the DFT software NRLMOL. In that NRLMOL is a general purpose DFT model for application to many different types of analysis, this tutoring is structured to put emphasis on those aspects and associated procedures of NRLMOL that are important for the numerical-simulation framework presented here.

1. fill in <name>.CLUSTER file
 - number of atoms
 - positions of atoms in atomic units/atomic numbers
2. run:
 - 000_prepare_relax <name>
 - creates directory <name>.RELAX

010_submit_relax <name>
submits relax
check GEOCNVRG for 'CONVERGE TRUE' message
020_copy_into_FREQ_subdir <name>
creates directory <name>.FREQ
030_generate_with_specsym <name>
040_prepare_first <name>
050_submit_first <name>
check for conversion:
'grep SELF nitro.specsym_1.stdout' should return 'SELF-CONSISTENCY REACHED'
060_prepare_all_the_rest
070_submit_all_the_rest
080_combine_FRCOUT
090_final_specsym
see specsym.out and infred.spc files
100_prepare_frozen_phonon
110_submit_frozen_phonon
120_combine_FRCOUT_frozen_phonon
130_specsym_frozen_phonon
see specsym.out.FPH

